```
> A := Matrix(3, 3,[[-55-7*x^2+22*x, -56-94*x^2+87*x, 97-62*
  x],
                [-83-73*x^2-4*x, -82-10*x^2+62*x, 71+80*x^2-44*x],
                [-10-17*x^2-75*x, 42-7*x^2-40*x, 75-50*x^2+23*x]]);
```

$$A := \begin{bmatrix} -7x^2+22x-55 & -94x^2+87x-56 & 97-62x \\ -73x^2-4x-83 & -10x^2+62x-82 & 80x^2-44x+71 \\ -17x^2-75x-10 & -7x^2-40x+42 & -50x^2+23x+75 \end{bmatrix} \quad (1)$$

```
> nrow := LinearAlgebra[RowDimension](A);;
  max_term_count := max(map(nops, A));
  max_degree := max(map(degree, A));
  max_coeff := max(map(coeffs, A));
  M_bound := 2 * nrow! * max_coeff^nrow * max_term_count^(nrow - 1)
  ;
  degree_bound := max_degree * nrow;
```

$$nrow := 3$$
$$max\_term\_count := 3$$
$$max\_degree := 2$$
$$max\_coeff := 97$$
$$M\_bound := 98568684$$
$$degree\_bound := 6 \quad (2)$$

```
> `mod` := mods;
  p := prevprime(101):
  primes := []:
  M := 1:
  crt_x := []:
  while M < M_bound do
      p := nextprime(p):
      primes := [op(primes), p];
      M := M*p;
      B := A mod p;
      xs := [];
      eval_xs := [];
      for i from 1 to (degree_bound + 1) do
          xs := [op(xs), i];
          C := eval(B,x=i) mod p:
          eval_xs := [op(eval_xs), Det(C) mod p];
      od;
      crt_x := [op(crt_x), interp(xs, eval_xs, x) mod p];

  od:



  print(primes);
  print(crt_x);
  result := collect(chrem(crt_x, primes),x);
```

$$mod := mods$$
$$[101, 103, 107, 109]$$
$$[31x^6-12x^5-39x^4+31x^3+24x^2-44x-42,\ 20x^6+50x^5+44x^4+x^3-20x^2-5x$$

$$-31,\ -4x^6+6x^5+44x^4-39x^3+13x^2-9x+10,\ 52x^6+9x^5+x^4-20x^3+25x^2+49x-49\Big]$$

$$result := 463520x^6-75964x^5-539985x^4+937816x^3-455486x^2+55203x-224262 \quad (3)$$

```
> maple_result := LinearAlgebra[Determinant](A);
```

$$maple\_result := 463520x^6-75964x^5-539985x^4+937816x^3-455486x^2+55203x-224262 \quad (4)$$

```
> expand(result - maple_result);
```

$$0 \quad (5)$$